91370
シート１に

```

    key
    1
    2
    3
    2
    1
    
```

とキーがかいてあって

シート２に

```

    key val
    1   abc
    2   xyz
    
```

という感じでキーとうめたい文字列がかかれてるとして

シート１の２列目にシート２のキーに対応する文字列を入れたいです

LOOKUPメソッドを使えばできるみたいなんですが  
`=LOOKUP(A2,'シート2'!A:A, 'シート2'!B:B)`  
という計算式で２列目うめたところ

```

    key val
    1   abc
    2   xyz
    3   xyz
    2   xyz
    1   abc
    
```

となぜか key=3 の存在しない値までうまってしまいます

存在しない値は空白のままにしたいのですが可能でしょうか？
CC BY-SA 4.0
2022-09-28T12:46:26.820
91358
2022-09-30T01:33:23.507
2022-09-28T12:52:32.467
question
google-spreadsheet
別シートを行検索して値を埋めたい
vlookupを使えばうまくいくと思います。※こちらの環境ではうまくいきました。

```

    =vlookup(A2, 'シート2'!A:B, 2, FALSE)
    
```

コードの重複が気になりますが、見つからなかったときに特定の文字列(今回はNG)とする場合は以下のように書けばよいです。

```

    =if(isna(vlookup(A2, 'シート2'!A:B, 2, FALSE)), "NG", vlookup(A2, 'シート2'!A:B, 2, FALSE))
    
```

* * *

以下追記です。  
`iferror`を使うともっと簡潔に書けました。

```

    =iferror(vlookup(A2, 'シート2'!A:B, 2, FALSE), "NG")
    
```

`filter`を使った別解です。

```

    =iferror(index(filter('シート2'!B:B,'シート2'!A:A=A2),1), "NG")
    
```

簡単に説明すると、'シート2'!A:Aの範囲でA2と同じ値を検索し、見つかった行のB列の値を返します。  
今回の質問では使う必要はありませんが、検索する列と値を返す列が連続していない場合にはこの方法が有効だと思います。
CC BY-SA 4.0
2022-09-29T08:15:23.007
91370
2022-09-30T01:33:23.507
2022-09-30T01:33:23.507
35558
35558
91358
answer